Teilchen im 1D Potential−Topf

In[1]:=
```
vpick = 16;(* pick the potential value *)
(* show allowed energy levels w as graphical solution *)
lhs[w_] := Sqrt[vpick / w^2 - 1];
rhs[w_] := Tan[w] UnitStep[Cos[w] Sin[w]] - Cot[w] UnitStep[-Cos[w] Sin[w]];
Plot[{lhs[w], rhs[w]}, {w, 0, Sqrt[vpick]}]
(* compute the number of allowed energy levels *)
n = 1 + Floor[Sqrt[4 vpick / Pi^2]]
(* compute the values of the allowed energy levels *)
wlist =
 Table[FindRoot[lhs[w] == rhs[w], {w, (i - 0.5) Pi / 2, (i - 1) Pi / 2, i Pi / 2}][[1, 2]], {i, n}]
(* plot the wave function and its modulus square for each energy *)
psi[y_, w_] := UnitStep[Sin[w] Cos[w]] Sqrt[w Sin[w] Cos[w]^2 / (Cos[w] + w Sin[w])]
    (UnitStep[y^2 - 1] Exp[w Tan[w] (1 - Abs[y])] + UnitStep[1 - y^2] Cos[w y] / Cos[w]) +
   UnitStep[-Sin[w] Cos[w]] Sqrt[w Cos[w] Sin[w]^2 / (w Cos[w] - Sin[w])]
    (UnitStep[y^2 - 1] Sign[y] Exp[-w Cot[w] (1 - Abs[y])] + UnitStep[1 - y^2] Sin[w y] / Sin[w]);
Plot[Table[psi[y, wlist[[i]]], {i, n}], {y, -2, 2}]
Plot[Table[psi[y, wlist[[i]]]^2, {i, n}], {y, -2, 2}]
(* show that the wave function is normalized to unity indeed *)
Table[Integrate[psi[y, wlist[[i]]]^2, {y, -Infinity, Infinity}], {i, 1, n}]
(* compute the probability that the particle is in the classically forbidden region *)
Table[1 - Integrate[psi[y, wlist[[i]]]^2, {y, -1, 1}], {i, 1, n}]
```

Out[4]=



Out[5]= 3

Out[6]= {1.25235, 2.47458, 3.5953}

Out[8]=



Out[9]=



Out[10]= {1., 1., 1.}

Out[11]= {0.0204264, 0.0923847, 0.293434}